
X04 – Input/Output Utilities

X04BCFP

NAG Parallel Library Routine Document

Note: before using this routine, please read the Users’ Note for your implementation to check for implementation-dependent
details. You are advised to enclose any calls to NAG Parallel Library routines between calls to Z01AAFP and Z01ABFP.

1 Description

X04BCFP reads an m by n real matrix As from an external file (stored in its natural, non-distributed
form) into an array in a cyclic two-dimensional block distribution on a logical grid of processors; As is a
submatrix of a larger mA by nA matrix A, i.e.,

As(1 : m, 1 : n) ≡ A(iA : iA + m − 1, jA : jA + n − 1).

Note: if iA = jA = 1, m = mA and n = nA, then As = A.

This routine distributes matrices in the form required by some routines in Chapter F07 and Chapter F08.

2 Specification

SUBROUTINE X04BCFP(NIN, M, N, A, IA, JA, IDESCA, IFAIL)
DOUBLE PRECISION A(∗)
INTEGER NIN, M, N, IA, JA, IDESCA(*), IFAIL

3 Usage
3.1 Definitions

The following definitions are used in describing the data distribution within this document:

mp – the number of rows in the Library Grid.
np – the number of columns in the Library Grid.
pr – the row grid coordinate of the calling processor.
pc – the column grid coordinate of the calling processor.
MX

b – the blocking factor for the distribution of the rows of a matrix X .
NX

b – the blocking factor for the distribution of the columns of a matrix X .
numroc(α,b�,q,s,k) – a function which gives the number of rows or columns of a distributed

matrix owned by the processor with the row or column coordinate q (pr

or pc), where α is the total number of rows or columns of the matrix,
b� is the blocking factor used (MX

b or NX
b ), s is the row or column

coordinate of the processor that possesses the first row or column of the
distributed matrix and k is either mp or np. The Library provides the
function Z01CAFP (NUMROC) for the evaluation of this function.

3.2 Global and Local Arguments

The following global input arguments must have the same value on entry to the routine on each processor
and the global output arguments will have the same value on exit from the routine on each processor:

Global input arguments: M, N, IA, JA, IFAIL, IDESCA(1), IDESCA(3:8)

Global output arguments: IFAIL

Note: NIN may have a different value on each processor but it is likely to be the same on each processor
– the only constraint is that it must be attached to the same file (or identical files) on each processor.

3.3 Distribution Strategy

On exit, the resulting array, A, contains the local portion of the matrix A distributed in a cyclic two-
dimensional block distribution for each processor. This data distribution is described in more detail in

the Essential Introduction of the NAG Parallel Library and in the F07 Chapter Introduction and the F08
Chapter Introduction. Array elements of A contained within the submatrix As will have been updated
with the values read from the input file.

4 Arguments

1: NIN — INTEGER Local Input

On entry: the unit number on which the external file is to be read.

Constraint: 0 < NIN ≤ 99.

2: M — INTEGER Global Input

On entry: m, the number of rows of the matrix As.

Constraint: 0 ≤ M ≤ IDESCA(3).

3: N — INTEGER Global Input

On entry: n, the number of columns of the matrix As .

Constraint: 0 ≤ N ≤ IDESCA(4).

4: A(∗) — DOUBLE PRECISION array Local Output

Note: array A is formally defined as a vector. However, you may find it more convenient to consider
A as a two-dimensional array of dimension (IDESCA(9),γ), where
γ ≥ numroc(JA+N−1,IDESCA(6),pc,IDESCA(8),np).

On exit: the local part of the matrix A, distributed in a cyclic two-dimensional block fashion.

5: IA — INTEGER Global Input

On entry: iA, the row index of A that identifies the first row of the submatrix As.

Constraint: 1 ≤ IA ≤ IDESCA(3) − M + 1.

6: JA — INTEGER Global Input

On entry: jA, the column index of A that identifies the first column of the submatrix As.

Constraint: 1 ≤ JA ≤ IDESCA(4) − N + 1.

7: IDESCA(∗) — INTEGER array Local Input

Note: the dimension of the array IDESCA must be at least 9.

Distribution: the array elements IDESCA(1) and IDESCA(3),. . .,IDESCA(8) must be global to
the processor grid and the elements IDESCA(2) and IDESCA(9) are local to each processor.

On entry: the description array for the matrix A. This array must contain details of the distribution
of the matrix A and the logical processor grid.

IDESCA(1), the descriptor type. For this routine, which uses a cyclic two-dimensional block
distribution, IDESCA(1) = 1;
IDESCA(2), the Library context, usually returned by a call to the Library Grid initialisation
routine Z01AAFP;
IDESCA(3), the number of rows, mA, of the matrix A;
IDESCA(4), the number of columns, nA, of the matrix A;
IDESCA(5), the blocking factor, MA

b , used to distribute the rows of the matrix A;
IDESCA(6), the blocking factor, NA

b , used to distribute the columns of the matrix A;
IDESCA(7), the processor row index over which the first row of the matrix A is distributed;
IDESCA(8), the processor column index over which the first column of the matrix A is
distributed;
IDESCA(9), the leading dimension of the conceptual two-dimensional array A.

Constraints:

IDESCA(1) = 1;
IDESCA(3) ≥ 0; IDESCA(4) ≥ 0;
IDESCA(5) ≥ 1; IDESCA(6) ≥ 1;
0 ≤ IDESCA(7) ≤ mp − 1; 0 ≤ IDESCA(8) ≤ np − 1;
IDESCA(9) ≥ max(1,numroc(IDESCA(3),IDESCA(5),pr,IDESCA(7),mp)).

8: IFAIL — INTEGER Global Input/Global Output

The NAG Parallel Library provides a mechanism, via the routine Z02EAFP, to reduce the amount
of parameter validation performed by this routine. For a full description refer to the Z02 Chapter
Introduction.

On entry: IFAIL must be set to 0, −1 or 1. For users not familiar with this argument (described in
the Essential Introduction) the recommended values are:

IFAIL = 0, if multigridding is not employed;
IFAIL = −1, if multigridding is employed.

On exit: IFAIL = 0 (or −9999 if reduced error checking is enabled) unless the routine detects an
error (see Section 5).

5 Errors and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output from the root processor (or processor
{0,0} when the root processor is not available) on the current error message unit (as defined by X04AAF).

5.1 Full Error Checking Mode Only

IFAIL = −2000

The routine has been called with a value of ICNTXT (stored in IDESCA(2)) which was not returned
by a call to Z01AAFP on one or more processors.

IFAIL = −1000

The logical processor grid and library mechanism (Library Grid) have not been correctly defined,
see Z01AAFP.

IFAIL < 0

On entry, one of the arguments was invalid:

if the kth argument is a scalar IFAIL = −k;
if the kth argument is an array and its jth element is invalid, IFAIL = −(100× k + j).

This error occurred either because a global argument did not have the same value on all logical
processors, or because its value on one or more processors was incorrect. An explanatory message
distinguishes between these two cases.

5.2 Any Error Checking Mode

IFAIL = 1

The routine has reached the end of the external file unexpectedly. This is probably due to supplying
values for M and N which are too large for the amount of data actually stored in the file.

IFAIL = 2

An error has occurred while reading the data, probably due to the routine trying to read an
incorrect data type. This may occur if there are non-numeric characters accidentally mixed with
the matrix data or if the current read position is not at the beginning of the matrix data when the
routine is called.

6 Further Comments

The routine needs exactly n matrix elements for each row of As and assumes that the next row of As

begins on the next line of the external file.

All of the processors read from the data file concurrently. On exit from the routine all processors will
have read data to the same point in the file. There is no way of checking that the data file is the same
for all processors; if the file is not the same for all processors, the behaviour of the routine could be
unpredictable.

7 References

None.

8 Example

This example reads in a matrix A from a data file, computes the LU factorization of the matrix, using
routine F07ADFP and prints the LU factors to standard output. The example uses a 2 by 2 logical
processor grid and a block size of 2.

Note: the listing of the Example Program presented below does not give a full pathname for the data file
being opened, but in general the user must give the full pathname in this and any other OPEN statement.

8.1 Example Text

* X04BCFP Example Program Text
* NAG Parallel Library Release 3 Revised. NAG Copyright 1999.
* .. Parameters ..

INTEGER NIN, NOUT
PARAMETER (NIN=5,NOUT=6)
INTEGER DT
PARAMETER (DT=1)
INTEGER MB, NB
PARAMETER (MB=2,NB=MB)
INTEGER MMAX, NMAX, LDA, IAROW, IACOL, NPMAX,MPMAX
PARAMETER (MMAX=8,NMAX=8,LDA=MMAX,IAROW=0,IACOL=0,NPMAX=2,

+ MPMAX=2)
INTEGER IW
PARAMETER (IW=NMAX)

* .. Local Scalars ..
INTEGER IA, ICNTXT, IFAIL, INFO, JA, M, N, NCOLS, NROWS
LOGICAL ROOT
CHARACTER*80 FORMAT

* .. Local Arrays ..
DOUBLE PRECISION A(LDA,NMAX), WORK(IW)
INTEGER IDESCA(9), IPIV(MMAX+MB)

* .. External Functions ..
LOGICAL Z01ACFP
EXTERNAL Z01ACFP

* .. External Subroutines ..
EXTERNAL F07ADFP, X04BCFP, X04BDFP, Z01AAFP, Z01ABFP

* .. Executable Statements ..
ROOT = Z01ACFP()
IF (ROOT) WRITE (NOUT,*) ’X04BCFP Example Program Results’

*
NROWS = NPMAX
NCOLS = MPMAX
IFAIL = 0

*
* Create Library Grid
*

CALL Z01AAFP(ICNTXT,NROWS,NCOLS,IFAIL)

OPEN (NIN,FILE=’x04bcfpe.d’)
* Skip heading in data file

READ (NIN,*)
READ (NIN,*) M, N, FORMAT

*
* Set the array descriptor of A
*

IA = 1
JA = 1
IDESCA(1) = DT
IDESCA(2) = ICNTXT
IDESCA(3) = MMAX
IDESCA(4) = NMAX
IDESCA(5) = MB
IDESCA(6) = NB
IDESCA(7) = IAROW
IDESCA(8) = IACOL
IDESCA(9) = LDA

*
IF (IA+M-1.LE.MMAX .AND. JA+N-1.LE.NMAX) THEN

*
* Read A from the data file
*

IFAIL = 0
CALL X04BCFP(NIN,M,N,A,IA,JA,IDESCA,IFAIL)

*
* Factorise the matrix
*

CALL F07ADFP(M,N,A,IA,JA,IDESCA,IPIV,INFO)
IF (INFO.EQ.0) THEN

*
* Print factor
*

IF (ROOT) WRITE (NOUT,*) ’Details of the’,
+ ’ factorization’

IFAIL = 0
CALL X04BDFP(NOUT,M,N,A,IA,JA,IDESCA,FORMAT,WORK,IFAIL)

ELSE IF (INFO.GT.0) THEN
IF (ROOT) WRITE (NOUT,*) ’Matrix is singular’

END IF
END IF

*
CLOSE (NIN)

*
* Close the Library Grid
*

IFAIL = 0
CALL Z01ABFP(ICNTXT,’N’,IFAIL)

*
STOP

*
END

8.2 Example Data

X04BCFP Example Program Data
4 4 ’(4F12.5)’ :Values of M, N and FORMAT
1.80 2.88 2.05 -0.89
5.25 -2.95 -0.95 -3.80
1.58 -2.69 -2.90 -1.04
-1.11 -0.66 -0.59 0.80 :End of matrix A

8.3 Example Results

X04BCFP Example Program Results
Details of the factorization

5.25000 -2.95000 -0.95000 -3.80000
0.34286 3.89143 2.37571 0.41286
0.30095 -0.46312 -1.51386 0.29482
-0.21143 -0.32988 0.00472 0.13137

